
   10  HOME 
       D$ = CHR$ (4)
       F$ = "textfile.txt"
   20  ONERR GOTO 100
   30  PRINT "Deleting file ";F$
       PRINT 
   40  PRINT D$;"delete ";F$
   50  PRINT "File ";F$;" deleted."
   60  GOTO 200
  100  ONERR GOTO 10000
  110  PRINT "File ";F$;" NOT deleted."
       PRINT 
  200  PRINT 
       PRINT "Creating file ";F$
  210  PRINT D$;"open ";F$
  220  PRINT D$;"write ";F$
  230  PRINT "This is line 1"
  240  PRINT "This is line 2"
  250  PRINT "This is line 3"
  260  PRINT "This is line 4"
  270  PRINT D$;"close ";F$
  280  PRINT 
       PRINT "File ";F$;" is now closed."
  300  PRINT 
       PRINT "Opening file ";F$;" to append."
  310  PRINT D$;"append ";F$
  320  PRINT D$;"write ";F$
  330  PRINT "This is an appended line."
  340  PRINT D$;"close ";F$
  350  PRINT 
       PRINT "File ";F$;" is now closed."
  400  PRINT 
       PRINT "Opening file ";F$;" to read."
       PRINT 
  405  ONERR GOTO 500
  410  PRINT D$;"open ";F$
  420  PRINT D$;"read ";F$
  440  INPUT A$
  450  PRINT A$
  460  GOTO 440
  500  ONERR GOTO 10000
  510  PRINT D$;"close ";F$
  520  PRINT 
       PRINT "File ";F$;" is now closed."
 9999  END 
10000  ONERR GOTO 10100
10010  PRINT 
       PRINT 
       PRINT "Final Error Recovery - cannot continue."
       PRINT 
10100  END 




