
   10  HOME 
       D$ = CHR$ (4)
       F$ = "TEXTFILE.TXT"
   15  A$ = "This is a line."
   20  ONERR GOTO 100
   30  PRINT "Deleting file ";F$
       PRINT 
   40  PRINT D$;"DELETE ";F$
   50  PRINT "File ";F$;" deleted."
   60  GOTO 200
  100  ONERR GOTO 10000
  110  PRINT "File ";F$;" NOT deleted."
       PRINT 
  200  PRINT 
       PRINT "Creating file ";F$
  210  PRINT D$;"OPEN ";F$
  220  PRINT D$;"WRITE ";F$
  230  FOR I = 1 TO 259
  240    FOR J = 1 TO 16
  250      PRINT A$
  260    NEXT 
       NEXT 
  270  PRINT D$;"CLOSE ";F$
  280  PRINT 
       PRINT "File ";F$;" is now closed."
  300  PRINT 
       PRINT "Opening file ";F$;" to append."
  310  PRINT D$;"OPEN ";F$
  320  PRINT D$;"APPEND ";F$
  325  PRINT D$;"WRITE ";F$
  330  PRINT "This is an appended line."
  340  PRINT D$;"CLOSE ";F$
  350  PRINT 
       PRINT "File ";F$;" is now closed."
  400  PRINT 
       PRINT "Opening file ";F$;" to read."
       PRINT 
  405  ONERR GOTO 500
  410  PRINT D$;"OPEN ";F$
  420  PRINT D$;"READ ";F$
  440  INPUT A$
  450  PRINT A$
  460  GOTO 440
  500  ONERR GOTO 10000
  510  PRINT D$;"CLOSE ";F$
  520  PRINT 
       PRINT "File ";F$;" is now closed."
 9999  END 
10000  ONERR GOTO 10100
10010  PRINT 
       PRINT 
       PRINT "Final Error Recovery - cannot continue."
       PRINT 
10100  END 




